Derived Types and Modules in Fortran

Victor Eijkhout, Susan Lindsey

Fall 2019

Types

Structures: `type`

- Fortran has structures too.
- Structures are a derived type: you can create variables of that type, but it's not a built-in type.
- Fortran derived types are declared in a `type` definition;
- the `type` keyword is also used for making structure variables.

Type declaration

`Type name` / `End Type` block.
Component declarations inside the block:

```
type mytype
  integer :: number
  character :: name
  real(4) :: value
end type mytype
```

Creating a type structure

Declare a type object in the main program:

```
Type(mytype) :: object1,object2
```

Initialize with type name:

```
object1 = mytype( 1, 'my_name', 3.7 )
```

Copying:

```
object2 = object1
```

Member access

Access structure members with %

```
Type(mytype) :: typed_object
typed_object%member = ....
```

Example

```
type point
   real :: x,y
end type point
```

```
type(point) :: p1,p2
p1 = point(2.5, 3.7)

p2 = p1
print *,p2%x,p2%y
```

Type definitions can go in the main program
(or use a `module`; see later)

Structures as subprogram argument

Structures can be passed as subprogram argument, just like any other datatype. This example:

- Takes a structure of `type(point)` as argument; and
- returns a `real(4)` result.
- The structure is declared as `intent(in)`.

Function with structure argument:

```
real(4) function length(p)
  implicit none
  type(point),intent(in) :: p
  length = sqrt( p%x**2 + p%y )
end function length
```

Function call

```
print *,"Length:",length(p2)
```

Exercise 1

Write a program that has the following:

- A type `Point` that contains real numbers `x,y`;
- a type `Rectangle` that contains two `Points`, corresponding to the lower left and upper right point;
- a function `area` that has one argument: a `Rectangle`.

Test your program.

Modules

Module definition

Modules look like a program, but without executable code:

```
Module definitions
  type point
     real :: x,y
  end type point
contains
  real(4) function length(p)
    implicit none
    type(point),intent(in) :: p
    length = sqrt( p%x**2 + p%y**2 )
  end function length
end Module definitions
```

Module use

Module imported through `use` statement;
comes before `implicit none`

Code:

```
Program size
  use definitions
  implicit none

  type(point) :: p1,p2
  p1 = point(2.5, 3.7)

  p2 = p1
  print *,p2%x,p2%y
  print *,"length:",length(p2)

end Program size
```

Output
[structf] typemod:

```
   2.50000000        3.70000005
length:    4.46542263
```

Exercise 2

Take exercise 1 and put all type definitions and all functions in a module.

Separate compilation of modules

Suppose program is split over two files:
`theprogram.F90` and `themodule.F90`.

- Compile the module: `ifort -c themodule.F90`; this gives
- an object file (extension: `.o`) that will be linked later, and
- a module file `modulename.mod`.
- Compile the main program:
  `ifort -c theprogram.F90` will read the `.mod` file; and finally
- Link the object files into an executable:
  `ifort -o myprogram theprogram.o themodule.o`
  The compiler is used as linker: there is no compiling in this step.

Important: the module needs to be compiled before any (sub)program that uses it.